"technicitate humana" by laice

[ Introduction ]

The story genre is "Dark Science Fiction". The release number is 01. The story headline is "from when we came, shall we return". The story description is "Human technology rapidly changes the course of the natural history of life in the universe. ". The story creation year is 2013.

[ Init]
The block giving rule is not listed in the check giving it to rules.
The block asking rule is not listed in the report asking it about rules.
The player is in the SymTech Tank.

[ Values ]
Power is a kind of value. The powers are max, medium, low and unpowered.
Intel-level is a kind of value. The intel-levels are operational, malfunctioning and broken.
Com-level is a kind of value. The com-levels are multimedia, video-only, sound-only, no-media.

[ Scenes ]

First Cycle is a scene. First Cycle begins when play begins. First Cycle ends when CZ5 is max.
Second Cycle is a scene. Second Cycle begins when First Cycle ends.

[ Regions ]
Station Epsilon is a region.

[ Rooms ]

A room has an indexed text called a name.

[ People ]

A person has an indexed text called forename. A person has an indexed text called fullname. A person has a number called health. The health of a person is usually 100.

A bot is a kind of person.



[ NPC's ]
CZ5 is a bot. CZ5 is in the Stasis Room.  CZ5 has a power. CZ5 is unpowered. CZ5 has an intel-level. CZ5 is operational. CZ5 has a com-level. CZ5 is multimedia. The fullname of CZ5 is "CZ5 - SymBot". The forename of CZ5 is "CZ5". The description of CZ5 is "A small rusted symbiotic assistance bot, used for menial tasks it's owner does not care to do."

[ Generic NPC Interactions ]

Asking someone about something is speech. 
Telling someone about something is speech. 
Answering someone that something is speech. 
Asking someone for something is speech. 

Check yourself giving a part to a bot:
	say "Giving...".

[ Powering up CZ5 ]

Carry out yourself giving a power cell to CZ5:
	now CZ5 is max.
	
Report yourself giving a power cell to CZ5:
	now the printed name of CZ5 is "CZ5";
	say "A bright flash sweeps the room as the broken bot comes to life!";
	say "A crisp monotone voice eminates from the bot, 'CZ5 - SymBot Diagnostic Complete : Operational'" instead.
	
[ CZ5 Interaction ]

Instead of speech when noun is CZ5:
	if CZ5 is not unpowered:
		if CZ5 is operational:
			if CZ5 is multimedia:
				repeat through Table of CZ5 dialog:
					if the topic understood includes topic entry:
						say "[dialog entry][paragraph break]";
						rule succeeds;					
				say "CZ5 bleeps and bloops.";
			otherwise :
				say "Unable to communicate.";
		otherwise :
			say "Not operational.";
	otherwise :
		say "Not powered."
	
	
Table of CZ5 Dialog
topic		dialog
"station epsilon"		"CZ5 says, 'Station Epsilon is where I do my laundry,' in monotone."
"symtech tank"		"CZ5 says, 'The rejuvenation and life support system of Station Epsilon.'"

[ Items ]
A part is a kind of thing. A part has a number called count. The description of a part is "It looks like it goes to something."

cpu ram assembly is a kind of part. videoboard is a kind of part. power cell is a kind of part.

shorted cpu ram assembly is a cpu ram assembly. overheated videoboard is a videoboard. leaking power cell is a power cell.

CZ5 is carrying a shorted cpu ram assembly. CZ5 is carrying an overheated videoboard.

[ Item locations ]
Leaking power cell is in the Stasis Room. 

[ Understanding]
Understand "a discarded bot" as CZ5 when CZ5 is proper-named.


[ First Cycle ]

When First Cycle begins:
	now right hand status line is "First Cycle";
	now the command prompt is "Login //> ";
	now the printed name of CZ5 is "a discarded bot";
	now CZ5 is proper-named.
	
To decide whether collecting names:
	if the command prompt is "Login //> ", yes;
	no.
	
Instead of looking when collecting names : do nothing.

Rule for printing the banner text when collecting names : do nothing.

Rule for constructing the status line when collecting names : do nothing.

	
Rule for printing a parser error when the latest parser error is the not a verb I recognise error:
	say "I don't get it." instead.
	

After reading a command when collecting names:
	now the fullname of yourself is "[the player's command]";
	now the forename of yourself is word number 1 in the fullname of yourself;
	now the command prompt is "//>";
	say "User set to [fullname of yourself].";
	reject the player's command.


[
Every turn:
	say "Name : [fullname of yourself] [line break]";
	say "Health : [health of yourself][line break]";
	say "CZ5 : [health of CZ5] [line break]";
	say "CZ5 Power : [power of CZ5] [line break]";
	say "CZ5 Intel : [intel-level of CZ5] [line break]";
	say "CZ5 Com : [com-level of CZ5] [line break]".
]


The SymTech Tank is a room. The name of SymTech Tank is "SymTech Tank".  The SymTech Tank is in Station Epsilon. "You find yourself immersed in liquid. A small display reads: [line break]Welcome to the Symbiosis Technologies SymTech(TM) tank version 7314 beta. You'll find your new model is equipped with  top-of-the-line features only available to our most loyal subscribers. "

Outside from the SymTech Tank is the Stasis Room. The name of Stasis Room is "Stasis Room." The Stasis Room is in Station Epsilon. "Dark metal boxes line the walls of an otherwise plain yet disconcerting room. It reminds you of a morgue. A cold green liquid lightly coats the floor, sending a chill through you with each footfall."

[ Second  Cycle ]

[ When Second Cycle begins:]
	